NAG Library Routine Document

G05LFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05LFF generates a vector of pseudorandom numbers taken from a gamma distribution with a and b.

2 Specification

```
SUBROUTINE G05LFF(A, B, N, X, IGEN, ISEED, IFAIL)
INTEGER           N, IGEN, ISEED(4), IFAIL
double precision  A, B, X(N)
```

3 Description

The gamma distribution has PDF (probability density function)

$$f(x) = \frac{1}{b^a \Gamma(a)} x^{a-1} e^{-x/b} \quad \text{if } 0 \leq x; \qquad a, b > 0.0$$

$$f(x) = 0 \qquad \text{otherwise.}$$

One of three algorithms is used to generate the variates depending upon the value of a:

(i) if $a < 1$, a switching algorithm described by Dagpunar (1988) (called G6) is used. The target distributions are $f_1(x) = cax^{a-1}/t^a$ and $f_2(x) = (1-c)e^{-(x-t)}$, where $c = t/\left(t + ae^{-t}\right)$, and the switching parameter, t, is taken as $1 - a$. This is Ahrens and Dieter's GS algorithm (see Ahrens and Dieter (1974)) in which $t = 1$;

(ii) if $a = 1$, the gamma distribution reduces to the exponential distribution and the method based on the logarithmic transformation of a uniform random variate is used;

(iii) if $a > 1$, the algorithm given by Best (1978) is used. This is based on using a Student's t-distribution with two degrees of freedom as the target distribution in an envelope rejection method.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05LFF.

4 References

Ahrens J H and Dieter U (1974) Computer methods for sampling from gamma, beta, Poisson and binomial distributions *Computing* **12** 223–46

Best D J (1978) Letter to the Editor *Appl. Statist.* **27** 181

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

1: A – ***double precision*** *Input*

On entry: a, the parameter of the gamma distribution.

Constraint: A > 0.0.

2: B – ***double precision*** *Input*

On entry: b, the parameter of the gamma distribution.

Constraint: B > 0.0.

3: N – INTEGER *Input*

On entry: n, the number of pseudorandom numbers to be generated.

Constraint: N ≥ 0.

4: X(N) – ***double precision*** array *Output*

On exit: the n pseudorandom numbers from the specified gamma distribution.

5: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

6: ISEED(4) – INTEGER array *Communication Array*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, A ≤ 0.0.

IFAIL $= 2$

On entry, B ≤ 0.0.

IFAIL $= 3$

On entry, N < 0.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints a set of five pseudorandom numbers from a gamma distribution with parameters $a = 5.0$ and $b = 1.0$, generated by a single call to G05LFF, after initialization by G05KBF.

9.1 Program Text

```
*     G05LFF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          N
      PARAMETER        (N=5)
*     .. Local Scalars ..
      DOUBLE PRECISION A, B
      INTEGER          IFAIL, IGEN, J
*     .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          ISEED(4)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05LFF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05LFF Example Program Results'
      WRITE (NOUT,*)
*     Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*
      A = 5.0D0
      B = 1.0D0
      IFAIL = 1
*
      CALL G05LFF(A,B,N,X,IGEN,ISEED,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
*
         WRITE (NOUT,99999) A, B
         WRITE (NOUT,99998) (X(J),J=1,N)
*
      ELSE
         WRITE (NOUT,99997) IFAIL
      END IF
*
99999 FORMAT (1X,'Gamma Dist --- A=',F3.1,', B=',F3.1)
99998 FORMAT (1X,F10.4)
99997 FORMAT (1X,' ** G05LFF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05LFF Example Program Results

 Gamma Dist --- A=5.0, B=1.0
     3.2806
     4.6512
     4.0683
     4.6252
     7.6745
```